Добрый день.
Кто подскажет почему данный запрос возвращает пустое значение.

```1c
    Сообщить(Контрагент);

    Запрос = Новый Запрос("
    |ВЫБРАТЬ
    | РеализацияТоваровУслуг.Номер Как Номер,
    | РеализацияТоваровУслуг.Дата Как Дата,
    | РеализацияТоваровУслуг.СуммаДокумента Как Сумма
    |ИЗ
    | Документ.РеализацияТоваровУслуг КАК РеализацияТоваровУслуг
    |ГДЕ
    | РеализацияТоваровУслуг.Контрагент = &Контрагент
    |И
    | РеализацияТоваровУслуг.Дата МЕЖДУ &ДатаНач И &ДатаКон
    |И
    | РеализацияТоваровУслуг.Проведен
    |И
    | НЕ РеализацияТоваровУслуг.ПометкаУдаления
    |");
    Запрос.УстановитьПараметр("ДатаНач", НачалоГода(ТекущаяДата()));
    Запрос.УстановитьПараметр("ДатаКон", КонецГода(ТекущаяДата()));
    Запрос.УстановитьПараметр("Контрагент", Контрагент);
    //
    Выборка = Запрос.Выполнить().Выбрать();
    Пока Выборка.Следующий() Цикл
    Сообщить(Выборка.Дата);
    Сообщить(Выборка.Номер);
    Сообщить(Выборка.Сумма);
    КонецЦикла;
```
---
Тогда либо сначала ссылку на контрагента получи, либо сделай поиск по наименованию типа
РеализацияТоваровУслуг.Контрагент.Наименование = &Контрагент

и на всякий случай - вместо
Запрос.УстановитьПараметр("Контрагент", Контрагент);
напиши
Запрос.УстановитьПараметр("Контрагент", СокрЛП(Контрагент));
---
Не покупал ничего у вас господин Рябов в этом году. Надобности не было.
---
Неудачная попытка пошутить
---
Федот, да не тот.
---
Значит Контрагент - не ссылка.

Всегда интересовало - зачем люди ставят условие на пометку удаления, если перед этим точно поставили условие на проведённые документы. Они правда думают, что у проведенных документов бывает пометка удаления?
---
Контрагент - не ссылка.

Расшифруйте пожалуйста
---
Наименование?
---
РеализацияТоваровУслуг.Ссылка.Контрагент = &Контрагент

так тоже возвращает пустое значение
---
сообщить(Строка(ТипЗнч(Контрагент))) - что выдаст?
---
сообщить(Строка(ТипЗнч(Контрагент)))
Строка
---
Ну, о чем я и говорил. Не ссылка.
---
Подскажите как исправить ?
---
Использовать ссылку контрагента, а не наименование. Например.
---
Тогда либо сначала ссылку на контрагента получи, либо сделай поиск по наименованию типа
РеализацияТоваровУслуг.Контрагент.Наименование = &Контрагент

и на всякий случай - вместо
Запрос.УстановитьПараметр("Контрагент", Контрагент);
напиши
Запрос.УстановитьПараметр("Контрагент", СокрЛП(Контрагент));
---
Что автоматически влечёт за собой левое соединение между таблицей документов РТУ и таблицей справочника Контрагенты...
---
:)
---
+(14) но лучше конечно по ссылке запрос строить, чтобы не было джоина по контрагенту
---
Ага, успел уточнить, пока я пальцами в телефон тыкал...
---
Если ссылки нет, то можно так попробовать:

КонтрагентСсылка = Справочники.Контрагенты.НайтиПоНаименованию(Контрагент,истина);
Если ЗначениеЗаполнено(КонтрагентСсылка) Тогда
    Запрос = Новый Запрос("
    |ВЫБРАТЬ
    | РеализацияТоваровУслуг.Номер Как Номер,
    | РеализацияТоваровУслуг.Дата Как Дата,
    | РеализацияТоваровУслуг.СуммаДокумента Как Сумма
    |ИЗ
    | Документ.РеализацияТоваровУслуг КАК РеализацияТоваровУслуг
    |ГДЕ
    | РеализацияТоваровУслуг.Контрагент = &Контрагент
    |И
    | РеализацияТоваровУслуг.Дата МЕЖДУ &ДатаНач И &ДатаКон
    |И
    | РеализацияТоваровУслуг.Проведен
    |");
    Запрос.УстановитьПараметр("ДатаНач", НачалоГода(ТекущаяДата()));
    Запрос.УстановитьПараметр("ДатаКон", КонецГода(ТекущаяДата()));
    Запрос.УстановитьПараметр("Контрагент", КонтрагентСсылка);
    //
    Выборка = Запрос.Выполнить().Выбрать();
    Пока Выборка.Следующий() Цикл
        Сообщить(Выборка.Дата);
        Сообщить(Выборка.Номер);
        Сообщить(Выборка.Сумма);
    КонецЦикла;
Иначе
    сообщить("не найден контрагент: "+Контрагент);
КонецЕсли;